Περισσότερη Prolog



Tamara de Ł empicka

Κωστής Σαγώνας <kostis@cs.ntua.gr> Νίκος Παπασπύρου <nickie@softlab.ntua.gr>

Αντικαταστάσεις (Substitutions)

• Μια *αντικατάσταση* είναι μια συνάρτηση που απεικονίζει μεταβλητές σε όρους:

$$\sigma = \{X \rightarrow a, Y \rightarrow f(a,b)\}$$

- Η παραπάνω αντικατάσταση σ αντιστοιχεί τη μεταβλητή **x** στο **a** και τη **y** στο **f** (**a**,**b**)
- Το αποτέλεσμα της εφαρμογής μιας αντικατάστασης σε έναν όρο δημιουργεί ένα στιγμιότυπο του όρου
 - Για παράδειγμα, $\sigma(\mathbf{g}(\mathbf{X},\mathbf{Y})) = \mathbf{g}(\mathbf{a},\mathbf{f}(\mathbf{a},\mathbf{b}))$
 - Ο όρος g(a, f(a,b)) είναι στιγμιότυπο του g(X,Y)

Περιεχόμενα

- Ενοποίηση
- Μοντέλα εκτέλεσης της Prolog
 - Το διαδικαστικό μοντέλο
 - Το μοντέλο υλοποίησης
 - Το αφηρημένο μοντέλο
- Περισσότερη Prolog
 - Κατηγορήματα εισόδου και εξόδου
 - Κατηγορήματα διαχείρισης της βάσης της Prolog
 - Αριθμητικοί υπολογισμοί και συγκρίσεις στην Prolog
- Παραδείγματα προγραμματισμού σε Prolog
- Αποχαιρετισμός στην Prolog

Περισσότερη Prolog

Ενοποίηση

- Δύο όροι της Prolog t_1 και t_2 ενοποιούνται εάν υπάρχει κάποια αντικατάσταση σ (ο ενοποιητής τους) που κάνει τους δύο όρους ακριβώς τους ίδιους: $\sigma(t_1) = \sigma(t_2)$
 - Οι όροι **a** και **b** δεν ενοποιούνται
 - Οι όροι $\mathbf{f}(\mathbf{X},\mathbf{b})$ και $\mathbf{f}(\mathbf{a},\mathbf{Y})$ ενοποιούνται: ένας ενοποιητής είναι ο $\{\mathbf{X} \to \mathbf{a}, \mathbf{Y} \to \mathbf{b}\}$
 - Οι όροι $\mathbf{f}(\mathbf{X},\mathbf{b})$ και $\mathbf{g}(\mathbf{Y},\mathbf{b})$ δεν ενοποιούνται
 - Οι όροι $\mathbf{a}(\mathbf{X}, \mathbf{X}, \mathbf{b})$ και $\mathbf{a}(\mathbf{b}, \mathbf{Y}, \mathbf{Y})$ ενοποιούνται: ένας ενοποιητής είναι ο $\{\mathbf{X} \to \mathbf{b}, \mathbf{Y} \to \mathbf{b}\}$
 - Οι όροι **a** (**X**, **X**, **b**) και **a** (**c**, **Y**, **Y**) δεν ενοποιούνται
 - Οι όροι $\mathbf{a}(\mathbf{X},\mathbf{f})$ και $\mathbf{a}(\mathbf{Y},\mathbf{f})$ ενοποιούνται: ένας ενοποιητής είναι ο $\{\mathbf{X} \to \mathbf{42},\,\mathbf{Y} \to \mathbf{42}\}$

Πολλαπλοί ενοποιητές

- parent(X,Y) KOU parent(fred,Z):
 - Ένας ενοποιητής είναι ο $\sigma_1 = \{X \rightarrow fred, Y \rightarrow Z\}$
 - Άλλος ένας είναι ο $\sigma_2 = \{X \rightarrow fred, Y \rightarrow mary, Z \rightarrow mary\}$
 - Άλλος ένας είναι ο σ_3 = {X ightarrow fred, Y ightarrow foo (42), Z ightarrow foo (42)}
- Η Prolog επιλέγει ενοποιητές όπως ο σ₁ οι οποίοι καθορίζουν ακριβώς τις αντικαταστάσεις που είναι αναγκαίες για την ενοποίηση των δύο όρων
- Με άλλα λόγια, η Prolog επιλέγει τον πιο γενικό ενοποιητή των δύο όρων (MGU – Most General Unifier)

Περισσότερη Prolog 5

Η Prolog χρησιμοποιεί ενοποίηση για τα πάντα

- Πέρασμα παραμέτρων
 - reverse([1,2,3],X)
- Δέσιμο μεταβλητών
 - x = 0
- Κατασκευή δεδομένων
 - X = .(1,[2,3])
- Επιλογή δεδομένων
 - -[1,2,3] = .(X,Y)

Ο πιο γενικός ενοποιητής (ΠΓΕ)

- Ο όρος t_1 είναι *πιο γενικός* από τον όρο t_2 εάν ο t_2 είναι στιγμιότυπο του t_1 αλλά ο t_1 δεν είναι στιγμιότυπο του t_2
 - Παράδειγμα: ο όρος parent(fred, Y) είναι πιο γενικός από τον όρο parent(fred, mary)
- Ένας ενοποιητής σ₁ δύο όρων t₁ και t₂ είναι *ο πιο γενικός* ενοποιητής εάν δεν υπάρχει άλλος ενοποιητής σ₂ τέτοιος ώστε ο όρος σ₂(t₁) να είναι πιο γενικός από τον όρο σ₁(t₁)
- Μπορεί να αποδειχθεί ότι ο πιο γενικός ενοποιητής είναι μοναδικός (αν αγνοήσουμε τα ονόματα των μεταβλητών)

Έλεγχος εμφάνισης (Occurs check)

Περισσότερη Prolog

- Μια μεταβλητή **x** και ένας όρος t ενοποιούνται με την
 - x και b ενοποιούνται: ο ΠΓΕ είναι $\{x \rightarrow b\}$

αντικατάσταση $\{\mathbf{x} \to t\}$:

- X KQI f(a,g(b)) EVOHOLOÚVTQI: O $\Pi F = \epsilon i \forall \alpha \in \{X \to f(a,g(b))\}$
- $X \text{ KQL } f(a, Y) \text{ EVOHOLOÚVTQL: O} \PiFE \text{ EÍVQL } \{X \rightarrow f(a, Y)\}$
- Εκτός εάν η μεταβλητή x περιλαμβάνεται στον όρο t:
 - Oι όροι $\mathbf X$ και $\mathbf f(\mathbf a,\mathbf g(\mathbf X))$ δεν ενοποιούνται: η αντικατάσταση $\{\mathbf X \to \mathbf f(\mathbf a,\mathbf g(\mathbf X))\}$ δεν είναι ενοποιητής
- Με άλλα λόγια, τουλάχιστον στη θεωρία, η ενοποίηση πρέπει να είναι καλά θεμελιωμένη (well-founded)

Ενοποίηση χωρίς έλεγχο εμφάνισης

 Η default ενοποίηση της Prolog δεν περιλαμβάνει έλεγχο εμφάνισης

```
append([], L, L).
append([H|L1], L2, [H|L3]) :-
append(L1, L2, L3).
```

```
?- append([], X, [a|X]).
X = [a|X].
?-
```

Αλλά το ISO Prolog standard περιλαμβάνει ένα
 κατηγόρημα με όνομα unify with occurs check/2

Μοντέλα Εκτέλεσης της Prolog

Περισσότερη Prolog

10

Περισσότερη Prolog

Το διαδικαστικό μοντέλο εκτέλεσης της Prolog

 Κάθε κατηγόρημα είναι μια διαδικασία για την απόδειξη στόχων

```
- p := q, r.
```

- Για την απόδειξη του στόχου p, πρώτα ενοποίησε το στόχο με την κεφαλή του κανόνα p, μετά απόδειξε το q, και μετά απόδειξε το r
- s.
 - Για την απόδειξη του στόχου s, ενοποίησε το στόχο με το s
- Η κάθε πρόταση (γεγονός ή κανόνας) αποτελεί ένα διαφορετικό τρόπο απόδειξης του στόχου
- Η απόδειξη μπορεί να περιλαμβάνει κλήσεις σε άλλα κατηγορήματα-διαδικασίες

Παραδείγματα

 Φανταστείτε τα κατηγορήματα της Prolog σαν συναρτήσεις σε ένα προστακτικό πρόγραμμα:

```
p :- q, r.
q :- s.
r :- s.
s.
```

```
boolean p() {return q() && r();}
boolean q() {return s();}
boolean r() {return s();}
boolean s() {return true;}
```

```
p:-p.
```

boolean p() {return p();}

Пερισσότερη Prolog 11 Περισσότερη Prolog 12

Οπισθοδρόμηση (Backtracking)

- Μια περιπλοκή: οπισθοδρόμηση
- Η Prolog κρατάει μια λίστα με όλους τους δυνατούς τρόπους με τους οποίους μπορεί να ικανοποιηθεί κάποιος στόχος και τους δοκιμάζει με τη σειρά μέχρι να ικανοποιήσει πλήρως το στόχο ή μέχρι να εξαντλήσει όλες τις δυνατότητες ικανοποίησής του
- Έστω ο στόχος ρ στο παρακάτω πρόγραμμα: ο στόχος ικανοποιείται αλλά μόνο με χρήση οπισθοδρόμησης

1. p:-q, r. 2. q:-s. 3. q. 4. r. 5. s:-0 = 1.

Περισσότερη Prolog

13

Αντικατάσταση

- Άλλη μια περιπλοκή: αντικατάσταση μεταβλητών
- Μια κρυμμένη ροή πληροφορίας


Η αντικατάσταση $\sigma_1 = \text{MGU}(\mathbf{p}(\mathbf{f}(\mathbf{Y})), t)$ εφαρμόζεται σε όλες τις μεθεπόμενες συνθήκες προς ικανοποίηση της πρότασης

Η αντικατάσταση σ_3 προκύπτει από την απόδειξη του όρου $\sigma_2(\sigma_1(\mathbf{r}(\mathbf{Y})))$

Η σύνθεση των αντικαταστάσεων επιστρέφεται στον καλούντα

Ο όρος που αποδείχθηκε είναι ο $\sigma_3(\sigma_2(\sigma_1(t)))$

14



Το μοντέλο υλοποίησης: Επίλυση (Resolution)

- Το βασικό βήμα συμπερασμού
- Κάθε κανόνας αναπαριστάται με μια λίστα από όρους (κάθε γεγονός αναπαριστάται από λίστα ενός στοιχείου)
- Κάθε βήμα επίλυσης χρησιμοποιεί μια από αυτές τις λίστες, μια φορά, για να επιτύχει κάποια πρόοδο στην απόδειξη μιας λίστας από όρους που είναι στόχοι προς απόδειξη για την απάντηση κάποιας ερώτησης

```
function resolution(clause, goals):

let sub = the MGU of head(clause) and head(goals)

return sub(body(clause) concatenated with tail(goals))
```

Παράδειγμα επίλυσης

```
Για την παρακάτω λίστα όρων προς απόδειξη: [p(x), s(x)]
```

και τον κανόνα:

Περισσότερη Prolog

p(f(Y)) :- q(Y), r(Y).

Ο ΠΓΕ είναι ο $\{x \to f(y)\}$, και στο επόμενο βήμα προκύπτει η λίστα αποτελούμενη από τους όρους:

```
resolution([p(f(Y)),q(Y),r(Y)], [p(X),s(X)])
= [q(Y),r(Y),s(f(Y))]
```

```
function resolution(clause, goals):
let sub = the MGU of head(clause) and head(goals)
return sub(body(clause) concatenated with tail(goals))
```

Περισσότερη Prolog 15 Περισσότερη Prolog 16

Ένας διερμηνέας της Prolog

```
function solve(goals)

if goals is empty then succeed()

else for each clause c in the program, in order

if head(c) does not unify with head(goals) then do nothing

else solve(resolution(c, goals))
```

Περισσότερη Prolog 17

Παραδείγματος συνέχεια

```
1. p(f(Y)) :-
q(Y),r(Y).
2. q(g(Z)).
3. q(h(Z)).
4. r(h(a)).
```

Μια μερική επίλυση του **p** (**X**), επεκταμένη:

Παράδειγμα

```
1. p(f(Y)) :-
	q(Y),r(Y).
2. q(g(Z)).
3. q(h(Z)).
4. r(h(a)).
```

Μια μερική επίλυση του **p (X)**:

- Η συνάρτηση solve δοκιμάζει τις τέσσερις προτάσεις τη μία μετά την άλλη
 - Η πρώτη πρόταση ταιριάζει, και η συνάρτηση solve καλεί τον εαυτό της αναδρομικά με το αποτέλεσμα της διάλυσης
 - Οι άλλες τρεις προτάσεις δεν ταιριάζουν: οι κεφαλές τους δεν ενοποιούνται με τον όρο της λίστας

```
Το παράδειγμα ολοκληρωμένο
```

```
1. p(f(Y)) :-
	q(Y),r(Y).
2. q(g(Z)).
3. q(h(Z)).
4. r(h(a)).
```

Περισσότερη Prolog

Η ολική επίλυση της ερώτησης **p (X)**:

18

20

```
solve([p(X)])
1. solve([q(Y),r(Y)])
     1. nothing
     2. solve([r(q(Z))])
          1. nothing
          2. nothing
          3. nothing
          4. nothing
     3. solve([r(h(Z))])
          1. nothing
          2. nothing
          3. nothing
          4. solve([]) -success!
     4. nothing
 2. nothing
 3. nothing
 4. nothing
```

Περισσότερη Prolog 19 Περισσότερη Prolog

Συλλογή των αντικαταστάσεων

```
function resolution(clause, goals, query):
    let sub = the MGU of head(clause) and head(goals)
    return (sub(tail(clause) concatenated with tail(goals)), sub(query))

function solve(goals, query)
    if goals is empty then succeed(query)
    else for each clause c in the program, in order
        if head(c) does not unify with head(goals) then do nothing
        else solve(resolution(c, goals, query))
```

- Τροποποιημένη συνάρτηση που δέχεται ως όρισμα την αρχική ερώτηση και εφαρμόζει όλες τις αντικαταστάσεις πάνω της
- Στο τέλος της επίλυσης, το αποδειχθέν στιγμιότυπο περνιέται ως όρισμα στη συνάρτηση succeed

Περισσότερη Prolog

```
    p(f(Y)):-
        q(Y),r(Y).
    q(g(Z)).
    q(h(Z)).
    r(h(a)).
```

```
Η ολική επίλυση της ερώτησης p (X):
```

```
solve([p(X)],p(X))
 1. solve([q(Y), r(Y)], p(f(Y)))
     1. nothing
     2. solve([r(g(Z))],p(f(g(Z))))
          1. nothing
          2. nothing
          3. nothing
          4. nothing
     3. solve([r(h(Z))],p(f(h(Z))))
          1. nothing
          2. nothing
          3. nothing
          4. solve([],p(f(h(a))))
     4. nothing
 2. nothing
 3. nothing
 4. nothing
```

Περισσότερη Prolog

22

Διερμηνείς της Prolog


- Ο διερμηνέας που μόλις είδαμε είναι λειτουργικός αλλά δεν χρησιμοποιείται από τις υλοποιήσεις της Prolog
- Όλες οι υλοποιήσεις της Prolog πρέπει μεν να λειτουργήσουν με τον τρόπο που μόλις περιγράψαμε αλλά συνήθως χρησιμοποιούν εντελώς διαφορετικές τεχνικές υλοποίησης: μεταφράζουν τον κώδικα σε γλώσσα κάποιας αφηρημένης μηχανής
- Η πιο συνηθισμένη τέτοια μηχανή είναι η Αφηρημένη Μηχανή του Warren (Warren Abstract Machine)

Το αφηρημένο μοντέλο: Δένδρα απόδειξης

- Θέλουμε να αναπαραστήσουμε με κάποιο τρόπο τη σειρά των λειτουργιών της εκτέλεσης, όμως χωρίς να περιορίζεται η τεχνική της υλοποίησης
- Τα δένδρα απόδειξης αποτελούν έναν τέτοιο φορμαλισμό:
 - Η ρίζα είναι η αρχική ερώτηση
 - Οι κόμβοι του δένδρου είναι λίστες από όρους προς απόδειξη, και κάθε κόμβος έχει ένα παιδί για κάθε πρόταση του προγράμματος


Περισσότερη Prolog 23 Περισσότερη Prolog 24

Παράδειγμα



Περισσότερη Prolog 25 Περισσότερη Prolog 26

Παράδειγμα απλοποιημένου δένδρου απόδειξης



Απλοποίηση των δένδρων απόδειξης

- Τα παιδιά κάθε κόμβου αντιστοιχούν στις προτάσεις του προγράμματος
- Και η σειρά τους είναι ίδια με τη σειρά εμφάνισής τους στο πρόγραμμα
- Μπορούμε να απλοποιήσουμε το δένδρο με το να απαλείψουμε όλους τους nothing κόμβους
- Οι κόμβοι αυτοί αντιστοιχούν σε προτάσεις οι οποίες δεν ενοποιούνται με το πρώτο στοιχείο της λίστας του κόμβου πατέρα

Σημασιολογία της Prolog

- Δοθέντος ενός προγράμματος και μιας ερώτησης, ένα σύστημα Prolog πρέπει να λειτουργήσει με τρόπο που καθορίζεται από μια πρώτα κατά βάθος, αριστερά προς τα δεξιά διάσχιση (depth-first, left-to-right traversal) του δένδρου απόδειξης
- Ένας τρόπος υλοποίησης είναι μέσω κάποιου διερμηνέα όπως αυτός ορίζεται από τη συνάρτηση solve
- Στην πράξη συνήθως χρησιμοποιούνται άλλοι, πιο αποδοτικοί, τρόποι υλοποίησης

Περισσότερη Prolog 27 Περισσότερη Prolog 28

Κατηγορήματα εισόδου και εξόδου

• Η είσοδος και η έξοδος όρων γίνεται σχετικά απλά

```
?- write('Hello world').
Hello world
true.
?- read(X).
|: hello(world).
X = hello(world).
?-
```

- Οποιοσδήποτε όρος της Prolog μπορεί να διαβαστεί
- Υπάρχει επίσης το κατηγόρημα n1/0 το οποίο είναι ισοδύναμο με την κλήση write('\n')

Περισσότερη Prolog

To κατηγόρημα retract/1

 Απομακρύνει την πρώτη πρόταση (κανόνα ή γεγονός) που ενοποιείται με το όρισμά του από τη βάση των δεδομένων

```
?- parent(joe,mary).
true.
?- retract(parent(joe,mary)).
true.
?- parent(joe,mary).
false.
```

• Υπάρχει επίσης και το κατηγόρημα retractall/1 το οποίο απομακρύνει όλες τις προτάσεις που ενοποιούνται με το όρισμά του

Το κατηγόρημα assert/1

 Προσθέτει ένα γεγονός ή έναν κανόνα στην εσωτερική βάση δεδομένων της Prolog (στο τέλος της βάσης)

```
?- dynamic parent/2.
true.
?- parent(X,Y).
false.
?- assert(parent(joe,mary)).
true.
?- parent(X,Y).
X = joe
Y = mary.
?-
```

Περισσότερη Prolog

30

Τα κατηγορήματα listing/0 και listing/1

- Τυπώνουν την εσωτερική βάση δεδομένων της Prolog:
 - όλη (listing/0) ή
 - μόνο κάποιο συγκεκριμένο κατηγόρημα (listing/1)

```
?- dynamic parent/2.
true.
?- assert(parent(joe,mary)).
true.
?- listing(parent/2).
:- dynamic parent/2.
parent(joe, mary).
true.
```

Περισσότερη Prolog 31 Περισσότερη Prolog 32

Μη αποτιμημένοι όροι

- Οι τελεστές της Prolog επιτρέπουν πιο συμπυκνωμένη γραφή των όρων, αλλά οι όροι δεν αποτιμώνται
- Όλες οι παρακάτω γραφές είναι ο ίδιος όρος Prolog:

```
1+ *(2,3)
+(1,2*3)
(1+(2*3))
1+2*3
```

Ο παραπάνω όρος δεν ενοποιείται με τον όρο 7

Περισσότερη Prolog

Η αποτίμηση προϋποθέτει τιμές...

```
?- Y = X + 2, X = 1.
Y = 1+2
X = 1.
?- X = 1, Y is X + 2.
X = 1
Y = 3.
?- Y is X + 2, X = 1.
ERROR: is/2: Arguments are not sufficiently instantiated
```

Αποτίμηση αριθμητικών εκφράσεων

```
?- X is 1 + 2 * 3.
X = 7.
```

- Το προκαθορισμένο κατηγόρημα is/2 μπορεί να χρησιμοποιηθεί για την αποτίμηση ενός όρου που είναι μια αριθμητική έκφραση
- is (X,Y) αποτιμά τον όρο Y και ενοποιεί τον όρο X με το αποτέλεσμα της αποτίμησης
- Συνήθως χρησιμοποιείται ως δυαδικός τελεστής

Περισσότερη Prolog

Αριθμητικές εκφράσεις και συναρτήσεις

- Σε έναν όρο **x** is **y**, οι υπό-όροι του **y** πρέπει να είναι αριθμοί ή *αποτιμώμενες συναρτήσεις*
- Οι συναρτήσεις αυτές περιλαμβάνουν τους
 προκαθορισμένους αριθμητικούς τελεστές +, -, * και /
- Όπως και προκαθορισμένες αριθμητικές συναρτήσεις,π.χ. abs (X), sqrt (X), floor (X), ...

Περισσότερη Prolog 35 Περισσότερη Prolog 36

Πραγματικές και ακέραιες τιμές

```
?- X is 1 / 2.

X = 0.5.

?- X is 4.0 / 2.0.

X = 2.0.

?- X is 5 // 2.

X = 2.

?- X is 4.0 / 2.

X = 2.0.
```

Δύο αριθμητικοί τύποι: ακέραιοι και πραγματικοί.

Οι περισσότερες αριθμητικοί τελεστές και συναρτήσεις είναι υπερφορτωμένοι για όλους τους συνδυασμούς ορισμάτων.

Η Prolog έχει δυναμικούς τύπους - οι τύποι χρησιμοποιούνται κατά το χρόνο εκτέλεσης για την επίλυση της υπερφόρτωσης.

Παρατηρείστε όμως ότι ο στόχος 2 = 2.0 αποτυγχάνει.

Αριθμητικές συγκρίσεις

• Τελεστές αριθμητικής σύγκρισης:

```
<, >, =<, >=, =:=, =\=
```

- Σε μια αριθμητική σύγκριση, η Prolog αποτιμά και τα δύο ορίσματα του τελεστή και συγκρίνει τις τιμές που προκύπτουν
- Άρα και τα δύο ορίσματα πρέπει να έχουν τιμές για όλες τις μεταβλητές τους

Περισσότερη Prolog

Περισσότερη Prolog

38

Αριθμητικές συγκρίσεις

```
?- 1+2 < 1*2.
false.
?- 1 < 2.0.
true.
?- 1+2 >= 1+3.
false.
?- X is 1-3, Y is 0-2.0, X =:= Y.
X = -2
Y = -2.0.
?-
```

Συγκρίσεις ισότητας στην Prolog

- Μέχρι στιγμής έχουμε χρησιμοποιήσει τρεις
 διαφορετικούς τελεστές σύγκρισης ισότητας:
 - X is Y αποτιμά τον όρο Y και ενοποιεί το αποτέλεσμα με το X π.χ. 3 is 1+2 επιτυγχάνει, αλλά 1+2 is 3 αποτυγχάνει
 - $\mathbf{X} = \mathbf{Y}$ ενοποιεί τους όρους \mathbf{X} και \mathbf{Y} , αλλά δεν τους αποτιμά π.χ. τόσο ο στόχος $\mathbf{3} = \mathbf{1} + \mathbf{2}$ όσο και ο $\mathbf{1} + \mathbf{2} = \mathbf{3}$ αποτυγχάνουν
 - x = := y αποτιμά τους δύο όρους και τους συγκρίνει αριθμητικά π.χ. τόσο ο στόχος 3 = := 1+2 και ο 1+2 = := 3 επιτυγχάνουν

Περισσότερη Prolog 39 Περισσότερη Prolog 40

Παραδείγματα

```
mylength1([],0).
mylength1([ |T], Len) :-
  mylength1(T, LenT),
  Len = LenT + 1.
```

```
mylength2([],0).
mylength2([ |T], Len) :-
  mylength2(T, LenT),
  Len is LenT + 1.
```

```
?- mylength1([a,b,c],X).
X = 0+1+1+1.
?- mylength1(X, 3).
... infinite loop ...^C
Action (h for help) ? a
% Execution aborted
```

```
?- mylength2([a,b,c],X).
x = 3.
?- mylength2(X,3).
X = [_, _, _] ;
... infinite loop ...
```

Παράδειγμα: sum

```
sum([], 0).
sum([Head|Tail],Sum) :-
 sum(Tail, TailSum),
 Sum is Head + TailSum.
```

```
?- sum([1, 2, 3], X).
X = 6.
?- sum([1, 2.5, 3], X).
X = 6.5.
?-
```

Περισσότερη Prolog

41

Περισσότερη Prolog

42

Παράδειγμα: gcd

```
gcd(X,Y,GCD): - \longleftarrow Προσοχή: όχι απλά
  X = := Y
  GCD is X.
gcd(X,Y,GCD):-
  X < Y
  NewY is Y - X,
  qcd(X, NewY, GCD).
gcd(X,Y,GCD):-
  X > Y
  NewX is X - Y,
  gcd (NewX,Y,GCD).
```

gcd(X,X,X)

Παράδειγμα: χρήση του κατηγορήματος ged

```
? - \gcd(5,5,X).
X = 5;
false.
?- \gcd(12,21,X).
x = 3;
false.
?- \gcd(91,105,X).
X = 7;
false.
?- gcd(91,21*5,7).
true ;
false.
?- \gcd(91, Y, 7).
ERROR: =:=/2: Arguments are not sufficiently instantiated
```

Παράδειγμα: factorial

```
factorial(X,Fact) :-
   X =:= 1, Fact is 1.
factorial(X,Fact) :-
   X > 1,
   NewX is X - 1,
   factorial(NewX,NF),
   Fact is X * NF.
```

```
?- factorial(5,F).
F = 120 ;
false.
?- factorial(2*5,F).
F = 3628800 ;
false.
?- factorial(-2,F).
false.
```

Διάζευξη και if-then-else στην Prolog

- Ο δυαδικός τελεστής ;/2 υλοποιεί τη διάζευξη (or)
- Ο ορισμός του είναι:

```
';'(X, _) :- X.
';'(_, Y) :- Y.
```

- Το if-then-else γράφεται με χρήση διάζευξης και του δυαδικού τελεστή ->/2
- Ο ορισμός του είναι:

```
(Cond -> Then ; _) :- Cond, Then.
(Cond -> _ ; Else) :- \+ Cond, Else.
```

Περισσότερη Prolog 45 Περισσότερη Prolog 46

Παράδειγμα χρήσης διάζευξης και if-then-else

```
factorial(X,Fact) :-
   X =:= 1, Fact is 1.
factorial(X,Fact) :-
   X > 1,
   NewX is X - 1,
   factorial(NewX,NF),
   Fact is X * NF.
```

```
factorial(X,Fact) :-
   ( X =:= 1, Fact is 1
   ; X > 1,
        NewX is X - 1,
        factorial(NewX,NF),
        Fact is X * NF
   ).
```

 \bigcirc

```
factorial(X,Fact) :-
  ( X =:= 1 -> Fact is 1
  ; X > 1,
    NewX is X - 1,
    factorial(NewX,NF),
    Fact is X * NF
  ).
```

Παράδειγμα χρήσης if-then-else

```
gcd(X,Y,GCD) :-
    X =:= Y,
    GCD is X.
gcd(X,Y,GCD) :-
    X < Y,
    NewY is Y - X,
    gcd(X,NewY,GCD).
gcd(X,Y,GCD) :-
    X > Y,
    NewX is X - Y,
    gcd(NewX,Y,GCD).
```

Από τον παραπάνω κώδικα μπορούμε να παραλείψουμε τον έλεγχο της συνθήκης **x** > **y** (και φυσικά το ->)

Περισσότερη Prolog 47 Περισσότερη Prolog 48

Συμπεριφορά του if-then χωρίς το else

Προσοχή: Ένα if-then χωρίς το else κομμάτι θα αποτύχει εάν η συνθήκη του if δεν είναι αληθής

 Αν θέλουμε να συνεχιστεί η εκτέλεση με τους στόχους μετά το if-then, πρέπει να χρησιμοποιήσουμε στο else κομμάτι το κατηγόρημα true/0

Δύο Παραδείγματα Προγραμμάτων





Περισσότερη Prolog

Περισσότερη Prolog

49

Προβλήματα αναζήτησης λύσης

- Η Prolog δε σχεδιάστηκε για προγραμματισμό αριθμητικών εφαρμογών (προφανώς)
- Τα προβλήματα στα οποία η Prolog δείχνει τις ικανότητές της είναι προβλήματα που χρησιμοποιούν αναζήτηση σε ένα χώρο λύσεων και όπου
 - Προσδιορίζουμε έναν ορισμό της λύσης με χρήση λογικής και
 - Αφήνουμε την Prolog να βρει αυτή τη λύση
- Θα εξετάσουμε τις λύσεις δύο προβλημάτων:
 - Το πρόβλημα του σακιδίου
 - Το πρόβλημα των οκτώ βασιλισσών

Το πρόβλημα του σακιδίου (Knapsack problem)

- Ετοιμάζουμε το σακίδιό μας για μια εκδρομή
- Στα ντουλάπια υπάρχουν τα εξής τρόφιμα:

Item	Weight in kilograms	Calories
bread	4	9200
pasta	2	4800
peanut butter	1	6700
baby food	3	6600

- Το σακίδιο χωράει πράγματα συνολικού βάρους 4 kg.
- Ποια επιλογή πραγμάτων βάρους ≤ 4 kg. μεγιστοποιεί το ποσό των θερμίδων;

Περισσότερη Prolog 51 Περισσότερη Prolog 52

Οι άπληστες μέθοδοι δε δουλεύουν

Item	Weight in kilograms	Calories
bread	4	9200
pasta	2	4800
peanut butter	1	6700
baby food	3	6600

- Πρώτα τις περισσότερες θερμίδες: bread = 9200
- Πρώτα τα πιο ελαφριά: peanut butter + pasta = 11500
- Βάσει του λόγου θερμίδες/βάρος: ίδιο με το παραπάνω
- (Η βέλτιστη επιλογή: peanut butter + baby food = 13300)
- (Κλασσικό πρόβλημα dynamic programming ⇒ algo, 7ο)

Περισσότερη Prolog 53

Αναπαράσταση

- Θα αναπαραστήσουμε κάθε είδος φαγητού με τον όρο food (Name, Weight, Calories)
- Το ντουλάπι του παραδείγματός μας αναπαρίσταται ως μια λίστα από τέτοιους όρους:

```
[food(bread,4,9200),
food(pasta,2,4800),
food(peanutButter,1,6700),
food(babyFood,3,6600)]
```

 Θα χρησιμοποιήσουμε την ίδια αναπαράσταση για τα περιεχόμενα του σακιδίου

Αναζήτηση

- Δεν υπάρχει γνωστός αλγόριθμος για το συγκεκριμένο πρόβλημα που
 - Πάντα δίνει τη βέλτιστη λύση για το πρόβλημα, και
 - Χρειάζεται λιγότερο από εκθετικό χρόνο για να τη βρει (συναρτήσει του πλήθους των τροφίμων)
- Κατά συνέπεια, δε θα πρέπει να ντρεπόμαστε αν χρησιμοποιήσουμε εξαντλητική αναζήτηση
- Το αντίθετο μάλιστα, επειδή η αναζήτηση είναι κάτι που η Prolog κάνει καλά

Περισσότερη Prolog 54

Ακολουθίες και υπακολουθίες

```
/*
   subseq(X, Y) succeeds when list X is the same as
   list Y, but with zero or more elements omitted.
   It can be used with any pattern of instantiations.
*/
subseq([], []).
subseq([Item | RestX], [Item | RestY]) :-
   subseq(RestX, RestY).
subseq(X, [_ | RestY]) :-
   subseq(X, RestY).
```

 Μια υπακολουθία μιας λίστας είναι ένα αντίγραφο της λίστας όπου κάποια στοιχεία της λίστας έχουν παραλειφθεί - στο παράδειγμά μας τα περιεχόμενα του σακιδίου είναι μια υπακολουθία του ντουλαπιού

Περισσότερη Prolog 55 Περισσότερη Prolog 56

Κάποια παραδείγματα

```
?- subseq([1,3],[1,2,3,4]).
true ;
false.
?- subseq(X,[1,2,3]).
X = [1, 2, 3];
X = [1, 2];
X = [1, 3];
X = [1];
X = [2, 3];
X = [2];
X = [3];
X = [];
false.
?-
```

Παρατηρείστε ότι το κατηγόρημα subseq/2 όχι μόνο μπορεί να ελέγξει κατά πόσο μια λίστα είναι μια υπακολουθία κάποιας άλλης αλλά μπορεί επίσης να παραγάγει υπακολουθίες, τις οποίες και θα χρησιμοποιήσουμε για τη λύση του προβλήματος του σακιδίου.

```
/*
   knapsackDecision(Items, Capacity, Goal, Knapsack)
   takes a list Items of food items, a positive number
   Capacity, and a positive number Goal. We unify
   Knapsack with a subsequence of Items representing
   a knapsack with total calories >= Goal, subject to
   the constraint that the total weight is =< Capacity.
*/
knapsackDecision(Items, Capacity, Goal, Knapsack) :-
   subseq(Knapsack, Items),
   weight(Knapsack, Weight),
   Weight =< Capacity,
   calories(Knapsack, Calories),
   Calories >= Goal.
```

Περισσότερη Prolog

59

Περισσότερη Prolog

Βοηθητικά κατηγορήματα

```
/*
  weight(Knapsack, Weight) given a Knapsack of food
  items returns their total Weight.
*/
weight(Knapsack, Weight) :-
  weight(Knapsack, 0, Weight).

weight([], W, W).
weight([food(_,W1,_) | Items], W, Weight) :-
  NW is W + W1,
  weight(Items, NW, Weight).
```

(Το κατηγόρημα calories/2 είναι αντίστοιχο.)

Για να δούμε που βρισκόμαστε...

- Η παραπάνω κλήση αποφασίζει εάν υπάρχει λύση που να ικανοποιεί το στόχο των ελάχιστων θερμίδων (10000)
- Όχι όμως υποχρεωτικά τη λύση που ζητάμε εμείς...

Περισσότερη Prolog 59 Περισσότερη Prolog 60

57

Αποφασισιμότητα και βελτιστοποίηση

- Μέχρι στιγμής λύσαμε το πρόβλημα της απόφασης για το πρόβλημα του σακιδίου
- Αυτό που θέλουμε να λύσουμε είναι το πρόβλημα της βελτιστοποίησης της λύσης του συγκεκριμένου προβλήματος
- Για να το επιτύχουμε, θα χρησιμοποιήσουμε ένα άλλο προκαθορισμένο κατηγόρημα της Prolog: findall/3

Περισσότερη Prolog

Συλλογή των απαντήσεων που θέλουμε

• Συνήθως, η πρώτη παράμετρος του **findall/3** είναι ένας όρος με τις μεταβλητές που υπάρχουν στο στόχο προς επίλυση

```
?- findall(X, subseq(X,[1,2]), L).
L = [[1, 2], [1], [2], []].
?-
```

 Η παραπάνω ερώτηση συλλέγει όλες τις απαντήσεις της ερώτησης-στόχου subseq (X, [1,2])

To κατηγόρημα findall/3

- findall(X, Goal, L)
 - Βρίσκει όλους τους τρόπους με τους οποίους ο στόχος Goal ικανοποιείται
 - Για τον καθένα από αυτούς, εφαρμόζει στον όρο **x** την ίδια αντικατάσταση με την οποία ο στόχος **Goal** ικανοποιήθηκε
 - Ενοποιεί τη λίστα τ με τη λίστα όλων αυτών των χ

```
?- findall(42, subseq(_,[1,2]), L).
L = [42, 42, 42, 42].
?-
```

- Υπάρχουν τέσσερις λύσεις για το subseq(_,[1,2])
- Συλλέξαμε μια λίστα από 42, ένα για κάθε λύση

Περισσότερη Prolog 62

```
/*
   knapsackOptimization(Items, Capacity, Knapsack) takes
   a list Items of food items and a positive integer
   Capacity. We unify Knapsack with a subsequence of
   Items representing a knapsack of maximum total
   calories, subject to the constraint that the total
   weight is =< Capacity.
*/
```

knapsackOptimization(Items, Capacity, Knapsack) :findall(K, legalKnapsack(Items, Capacity, K), L),
maxCalories(L, Knapsack).

 Περισσότερη Prolog
 63

```
/*
  legalKnapsack(Items,Capacity,Knapsack) takes a list
  Items of food items and a positive number Capacity.
  We unify Knapsack with a subsequence of Items whose
  total weight is =< Capacity.

*/
legalKnapsack(Items, Capacity, Knapsack):-
  subseq(Knapsack, Items),
  weight(Knapsack, W),
  W =< Capacity.</pre>
```

```
maxCalories(List, Result) takes a non-empty List of
 lists of food items. We unify Result with an element
 from the list that maximizes the total calories.
 We use a helper predicate maxC that takes four
 parameters: the remaining list of lists of food
 items, the best list of food items seen so far, its
  total calories, and the final result.
maxCalories([First | Rest], Result) :-
  calories(First, FirstC),
 maxC(Rest, First, FirstC, Result).
maxC([], Sofar, , Sofar).
maxC([First | Rest], , MC, Result) :-
  calories(First, FirstC),
 MC =< FirstC,</pre>
 maxC(Rest, First, FirstC, Result).
maxC([First | Rest], Sofar, MC, Result) :-
  calories(First, FirstC),
 MC > FirstC,
 maxC(Rest, Sofar, MC, Result).
```

Περισσότερη Prolog 65

```
Περισσότερη Prolog 66
```

```
maxCalories(List, Result) takes a non-empty List of
  lists of food items. We unify Result with an element
  from the list that maximizes the total calories.
  We use a helper predicate maxC that takes four
  parameters: the remaining list of lists of food
  items, the best list of food items seen so far, its
  total calories, and the final result.
maxCalories([First | Rest], Result) :-
  calories(First, FirstC),
  maxC(Rest, First, FirstC, Result).
maxC([], Sofar, , Sofar).
maxC([First | Rest], Sofar, MC, Result) :-
  calories(First, FirstC),
  ( MC =< FirstC ->
      NSofar = First, NMC = FirstC
                                              Το ίδιο με χρήση
                                              if-then-else
      NSofar = Sofar, NMC = MC
  ),
  maxC(Rest, NSofar, NMC, Result).
```


Περισσότερη Prolog 67 Περισσότερη Prolog

Το πρόβλημα των οκτώ βασιλισσών (8-queens)

- Κάποιες πληροφορίες για το σκάκι:
 - Παίζεται σ' ένα τετράγωνο 8×8
 - Μια βασίλισσα μπορεί να κινηθεί οριζόντια, κάθετα ή διαγώνια για όσα τετράγωνα θελήσει

 Δύο βασίλισσες απειλούν η μία την άλλη εάν είναι στην ίδια γραμμή, στήλη ή διαγώνιο (και η μία μπορεί να κινηθεί άμεσα στο τετράγωνο της άλλης)


 Το πρόβλημα: βρες ένα τρόπο να τοποθετηθούν οκτώ βασίλισσες σε μια κενή σκακιέρα έτσι ώστε καμία βασίλισσα να μην απειλείται από κάποια άλλη



Περισσότερη Prolog 69

Παράδειγμα

- Ένας σχηματισμός σκακιέρας είναι μια λίστα από βασίλισσες
- Ο παρακάτω είναι για τις βασίλισσες [2/5,3/7,6/1]



Αναπαράσταση

- Θα μπορούσαμε να αναπαραστήσουμε τη βασίλισσα στη στήλη 2, σειρά 5 με τον όρο queen (2,5)
- Αλλά αφού δεν υπάρχουν άλλα κομμάτια στη σκακιέρα—
 π.χ. δε θα έχουμε κάποιο pawn (X,Y) ή king (X,Y)—θα
 χρησιμοποιήσουμε απλώς έναν όρο της μορφής X/Y
- (Δε θα τον αποτιμήσουμε ως διαίρεση)

Περισσότερη Prolog 70

```
nocheck(L, X/Y) takes a queen X/Y and a list
 of queens. It succeeds if and only if the X/Y
 queen holds none of the others in check.
nocheck([], ).
nocheck([X1/Y1 | Rest], X/Y) :-
 X = \ X1
 Y = \ Y1
  abs(Y1-Y) = = abs(X1-X),
  nocheck (Rest, X/Y).
 legal(L) succeeds if L is a legal placement of
  queens: all coordinates in range and no queen
  in check.
legal([]).
legal([X/Y | Rest]) :-
 legal (Rest),
 member(X, [1,2,3,4,5,6,7,8]),
 member(Y, [1,2,3,4,5,6,7,8]),
 nocheck (Rest, X/Y).
                                                  72
```

Περισσότερη Prolog 71

Επάρκεια

 Έχουμε ήδη αρκετά συστατικά για να λύσουμε το πρόβλημα: η ερώτηση legal (X) βρίσκει όλους τους επιτρεπτούς σχηματισμούς:

```
?- legal(X).
X = [] ;
X = [1/1] ;
X = [1/2] ;
X = [1/3]
```

Περισσότερη Prolog

Υπάρχει δυνατότητα για βελτίωση;

- Η εύρεση της λύσης με αυτόν τον τρόπο παίρνει πολύ χρόνο
- Στη συνέχεια, η Prolog βρίσκει απλές αναδιατάξεις της πρώτης λύσης:

```
?- X = [_,_,_,_,], legal(X).

X = [8/4, 7/2, 6/7, 5/3, 4/6, 3/8, 2/5, 1/1];

X = [7/2, 8/4, 6/7, 5/3, 4/6, 3/8, 2/5, 1/1];

X = [8/4, 6/7, 7/2, 5/3, 4/6, 3/8, 2/5, 1/1];

X = [6/7, 8/4, 7/2, 5/3, 4/6, 3/8, 2/5, 1/1]
```


Λύση για τις οκτώ βασίλισσες

 Φυσικά, η παραπάνω «λύση» θα πάρει πολύ χρόνο: θα αρχίσει με το να βρει όλες τις 64 λύσεις με μια βασίλισσα, και μετά θα αρχίσει με όλες τις λύσεις με δύο, κοκ.

 Μπορούμε να «εκβιάσουμε» μια λύση με 8 βασίλισσες με την ερώτηση:

```
?- X = [_,_,_,_,_,_],
| legal(X).

X = [8/4, 7/2, 6/7, 5/3,
4/6, 3/8, 2/5, 1/1]
?-
```



Περισσότερη Prolog

Περισσότερη Prolog

Βελτίωση του προγράμματος

- Προφανώς κάθε λύση έχει μία βασίλισσα σε κάθε στήλη
- Άρα, κάθε λύση μπορεί να γραφεί με τη μορφή:

$$X = [1/_, 2/_, 3/_, 4/_, 5/_, 6/_, 7/_, 8/_]$$

 Δίνοντας έναν όρο αυτής της μορφής περιορίζουμε την αναζήτηση, επιταχύνοντάς την, και αποφεύγουμε την εύρεση απλών αναδιατάξεων

```
/* eightqueens(X) succeeds if X is a legal
   placement of eight queens, listed in order
   of their X coordinates.

*/
eightqueens(X) :-
   X = [1/_,2/_,3/_,4/_,5/_,6/_,7/_,8/_],
   legal(X).
```

Περισσότερες βελτιώσεις

 Επειδή ξέρουμε ότι όλες οι Χ-συντεταγμένες είναι εντός του διαστήματος και διαφορετικές μεταξύ τους, το πρόγραμμα μπορεί λιγάκι να βελτιωθεί

```
nocheck([], _).
nocheck([X1/Y1 | Rest], X/Y) :-
% X =\= X1, assume the X's are distinct
Y =\= Y1,
abs(Y1-Y) =\= abs(X1-X),
nocheck(Rest, X/Y).

legal([]).
legal([]).
legal([X/Y | Rest]) :-
legal(Rest),
% member(X, [1,2,3,4,5,6,7,8]), assume X in range
member(Y, [1,2,3,4,5,6,7,8]),
nocheck(Rest, X/Y).
```

Περισσότερη Prolog

77

Ένα πείραμα

```
legal([]).
legal([X/Y | Rest]) :-
  legal(Rest),
  % member(X, [1,2,3,4,5,6,7,8]), assume X in range
  1 =< Y, Y =< 8, % was member(Y,[1,2,3,4,5,6,7,8]),
  nocheck(Rest, X/Y).</pre>
```

- Το παραπάνω κατηγόρημα δε δουλεύει
- Εγείρει εξαίρεση: "arguments not sufficiently instantiated"
- Η κλήση του κατηγορήματος member/2 δεν είναι απλώς κάποιος έλεγχος ότι οι συντεταγμένες είναι εντός ορίων αλλά είναι παραγωγή των συντεταγμένων

Βελτιωμένη λύση στο πρόβλημα

- Το πρόγραμμα τώρα τρέχει αρκετά πιο γρήγορα
- Για παράδειγμα, δε βρίσκει πλέον όλες τις αναδιατάξεις

```
?- eightqueens(X).
X = [1/4, 2/2, 3/7, 4/3, 5/6, 6/8, 7/5, 8/1];
X = [1/5, 2/2, 3/4, 4/7, 5/3, 6/8, 7/6, 8/1]
```

Περισσότερη Prolog 78

Άλλο ένα πείραμα

```
legal([]).
legal([X/Y | Rest]) :-
   % member(X, [1,2,3,4,5,6,7,8]), assume X in range
   member(Y, [1,2,3,4,5,6,7,8]),
   nocheck(Rest, X/Y),
   legal(Rest). % formerly the first condition
```

- Εγείρει εξαίρεση: "arguments not sufficiently instantiated"
- Η κλήση legal (Rest) πρέπει να προηγείται διότι παράγει την μερική λύση που χρειάζεται η αναδρομική κλήση του nocheck

Περισσότερη Prolog 79 Περισσότερη Prolog 80

Εύρεση μίας μόνο λύσης

 Αν θέλαμε να βρούμε μία μόνο λύση του προβλήματος θα μπορούσαμε να χρησιμοποιήσουμε οποιοδήποτε από τους δύο παρακάτω ορισμούς του σχετικού κατηγορήματος

```
/*
    eightqueens1(X) finds
    one solution to the
    eight queens problem.
*/
eightqueens1(X) :-
    once(eightqueens(X)).
```

```
/*
   eightqueens1(X) finds
   one solution to the
   eight queens problem.
*/
eightqueens1(X) :-
   eightqueens(X),
  !.
```

Περισσότερη Prolog 81 Περισσότερη Prolog 82

Παράδειγμα: Οπισθοδρόμηση χωρίς και με cut

```
p(X,Y):-
x(X),
y(Y).
p(X,Y):-
z(X,Y).
p(4,d).
x(1).
x(2).
y(a).
y(b).
z(3,c).
```

```
?- p(X,Y).

X = 1
Y = a;

X = 1
Y = b;

X = 2
Y = a;

X = 2
Y = c;

X = 3
Y = c;

X = 4
Y = d;

false.
```

```
p(X,Y) : -
             ?-p(X,Y).
  x(X),
             X = 1
  !,
             Y = a;
  y(Y).
p(X,Y) : -
            X = 1
  z(X,Y).
            Y = b;
p(4,d).
             false.
x(1).
x(2).
y(a).
y(b).
```

z(3,c).

To cut (!) της Prolog

- Το προκαθορισμένο κατηγόρημα !/0 (διαβάζεται cut)
 χρησιμοποιείται για να περιορίσει την οπισθοδρόμηση
- Η εκτέλεσή του πάντα επιτυγχάνει και «κλαδεύει»:
 - Όλους τους εναλλακτικούς τρόπους ικανοποίησης των υποστόχων που πιθανώς να υπάρχουν μεταξύ της κεφαλής ενός κανόνα και του σημείου που το! εμφανίζεται στον κανόνα, και
 - Όλους τους κανόνες του ίδιου κατηγορήματος που μπορεί να έπονται του κανόνα που περιέχει το !
- Η χρήση του πρέπει να γίνεται με φειδώ και ύστερα από αρκετή σκέψη

Μέρη της Prolog που δεν εξετάσαμε

- Τη δυνατότητα ορισμού νέων τελεστών
- Το χειρισμό εξαιρέσεων
 - Εξαιρέσεις που εγείρονται από το σύστημα και από το χρήστη
 - Τα σχετικά κατηγορήματα throw και catch
- Τις βιβλιοθήκες και πολλά από τα προκαθορισμένα κατηγορήματα της γλώσσας

Περισσότερη Prolog 83 Περισσότερη Prolog 84

Αποχαιρετισμός στην Prolog

- Δε χρειάστηκε να παραλείψουμε τόσο ποσοστό της Prolog όσο παραλείψαμε στην ML και στη Java
- Η Prolog είναι μια μικρή γλώσσα
- Επίσης, είναι ένα αρκετά ισχυρό εργαλείο επίλυσης
 (κάποιου είδους) προβλημάτων αλλά ο επιδέξιος χειρισμός
 της από τον προγραμματιστή δεν είναι πολύ εύκολος
- Ο επιδέξιος χειρισμός της Prolog προϋποθέτει την υιοθέτηση και εξοικείωση με μια διαφορετική φιλοσοφία προγραμματισμού

Περισσότερη Prolog